Mat 1. 2-timersprøve den 14. maj 2019.

JE 13.5.19, rev KS 12.4.23

Opgave 1

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  vop:=proc(X) op(convert(X,list)) end proc:
  grad:=X->convert(Student[VectorCalculus][Del](X),Vector):
> with(LinearAlgebra):
```

En reel funktion f af to reelle variable er givet ved

```
> f:=(x,y)->y/exp(x);
```

$$f := (x, y) \mapsto \frac{y}{e^x}$$

```
> f(x,y);
```

$$\frac{y}{e^x}$$

hvor $(x, y) \in \mathbb{R}^2$.

Spørgsmål 1

Niveaukurven svarende til niveauet 0 er givet ved

$f(x, y) = \frac{y}{e^x} = 0 \Leftrightarrow y = 0$ og $x \in \mathbb{R}$, hvilket er hele x-aksen.

Nievaukurven svarende til niveauet 1 er givet ved

$f(x, y) = \frac{y}{e^x} = 1 \Leftrightarrow y = e^x$, $x \in \mathbb{R}$, hvilket er grafen for eksponentialfunktionen.

```
> contourplot(f(x,y),x=-1..1,y=-1..2,contours=[0,1]);
```



Spørgsmål 2

```
> diff(f(x,y),x);diff(f(x,y),y);
```

$$-\frac{y}{e^x}$$

$$\frac{1}{e^x}$$

$\nabla f(x, y) = (f'_x(x, y), f'_y(x, y)) = \left(-\frac{y}{e^x}, \frac{1}{e^x}\right)$.

$A = (0, 1)$, $B = (1, 2)$, $\mathbf{v} = \overrightarrow{AB} = (1, 1)$, $\mathbf{e} = \frac{\mathbf{v}}{|\mathbf{v}|} = \left(\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right)$, $\nabla f(A) = (-1, 1)$.

Den retningsafledede af f i punktet A i retningen mod punktet B er da
$f'(A; \mathbf{e}) = \mathbf{e} \cdot \nabla f(A) = 0$.
Da gradienten i et punkt står vinkelret på niveaukurven gennem punktet, så er resultatet ikke overraskende, da $\mathbf{e}$ er parallel med 1-niveaukurvens tangent i A.
Lad $\mathbf{e}$ være en enhedsvektor ud fra punktet A. Den retningsafledede af f i punktet A i retningen af $\mathbf{e}$ er

$f'(A;\mathbf{e}) = \mathbf{e}\cdot\nabla f(A) = |\nabla f(A)|\cos(\varphi) = \sqrt{2}\cos(\varphi)$, hvor φ er vinklen mellem **e** og $\nabla f(A)$.
For $\varphi = 0$ svarende til at **e** er i gradientens retning fås $\sqrt{2}$, som er den største værdi den retningsafledede af f i punktet A kan opnå.

En kurve K i (x, y)-planen er givet ved parameterfremstillingen

```
> r:=u-><u,1+u-u^2>:
> r(u);
```

$$\begin{bmatrix} u \\ -u^2+u+1 \end{bmatrix}$$

hvor $u \in \mathbb{R}$.

Spørgsmål 3

```
> ru:=diff(r(u),u);
```

$$ru := \begin{bmatrix} 1 \\ -2u+1 \end{bmatrix}$$

```
> gradf:=unapply(<diff(f(x,y),x),diff(f(x,y),y)>,[x,y]):
```

Gradienten for f i kurvepunktet $\mathbf{r}(u)$ er

```
> gradf(vop(r(u)));
```

$$\begin{bmatrix} -\frac{-u^2+u+1}{e^u} \\ \frac{1}{e^u} \end{bmatrix}$$

```
> g:=prik(gradf(vop(r(u))),ru);
```

$$g := -\frac{-u^2+u+1}{e^u} + \frac{-2u+1}{e^u}$$

```
> solve(g=0,u);
```

$$0, 3$$

De punkter på kurven K hvori gradienten for f og tangentvektoren $\mathbf{r}'(u)$ er ortogonale er således punkterne $\mathbf{r}(0) = (0, 1)$ og $\mathbf{r}(3) = (3, -5)$.
Situationen i punktet (0, 1):

```
> P1:=plot(exp(x),x=-1..4,color=red):
> P2:=plot(1+x-x^2,x=-1..4,color=blue):
> P3:=plot(x+1,x=-1..1):
> display(P1,P2,P3,scaling=constrained,view=-5..3);
```



1-niveaukurven og kurven K tangerer hinanden i punktet A = (0, 1).

Opgave 2

```
> restart:with(plots):with(LinearAlgebra):with(student):
```

Det oplyses, at den symmetriske matrix

```
> A:=<<2,-1>|<-1,2>>;
```

$$A := \begin{bmatrix} 2 & -1 \\ -1 & 2 \end{bmatrix}$$

har egenværdierne 1 og 3 med tilhørende egenrum E_1 = span{(1, 1)} og E_3 = span{(-1, 1)}.
Da **A** er symmetrisk, er de to egenrum E_1 og E_3 ortogonale.

Et andengradspolynomium i to reelle variable er givet ved

```
> f:=(x,y)->2*x^2-2*x*y+2*y^2-4*x+2*y+2:
> f(x,y);
```

$$2\,x^2 - 2\,y\,x + 2\,y^2 - 4\,x + 2\,y + 2$$

Spørgsmål 1

På matrixform haves

$$f(x, y) = \begin{bmatrix} x & y \end{bmatrix} \begin{bmatrix} 2 & -1 \\ -1 & 2 \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix} + \begin{bmatrix} -4 & 2 \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix} + 2.$$

Matricen for den indgående kvadratiske form

$$P(x, y) = \begin{bmatrix} x & y \end{bmatrix} \begin{bmatrix} 2 & -1 \\ -1 & 2 \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix}$$

er netop den givne symmetriske matrix **A**.

Af det oplyste følger så, at

```
> q1:=<1/sqrt(2),1/sqrt(2)>;
```

$$q1 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \end{bmatrix}$$

er en ortonormal basis for E_1 og at

```
> q2:=<-1/sqrt(2),1/sqrt(2)>;
```

$$q2 := \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \end{bmatrix}$$

hvor $\mathbf{q}2 = \widehat{\mathbf{q}1}$ er en ortonormal basis for E_3.

Egenvektorerne ($\mathbf{q}1$, $\mathbf{q}2$) er da en ortonormal basis for $\mathbb{R}^2$ udstyret med det sædvanlige skalarprodukt.

Hvis der i planen indføres et sædvanligt retvinklet koordinatsystem (O; **i**, **j**) så er (O; **q**1, **q**2) et nyt sædvanligt retvinklet koordinatsystem i planen fremkommet ved en drejning af det givne koordinatsystem 45^{o} om O mod uret.

Sættes

```
> Q:=<q1|q2>;
```

$$Q := \begin{bmatrix} \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \end{bmatrix}$$

som er basisskiftematricen, der i $\mathbb{R}^2$ skifter fra q-koordinater til standard e-koordinater og

```
> Lambda:=DiagonalMatrix([1,3]);
```

$$\Lambda := \begin{bmatrix} 1 & 0 \\ 0 & 3 \end{bmatrix}$$

så er **Q** positiv ortogonal og

$$\mathbf{Q}^{-1}\,\mathbf{A}\,\mathbf{Q} = \mathbf{Q}^{\mathrm{T}}\,\mathbf{A}\,\mathbf{Q} = \mathbf{\Lambda}$$

Ved anvendelse af den positive ortogonale substitution

$$\begin{bmatrix} x \\ y \end{bmatrix} = \mathbf{Q} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix},$$

som angiver sammenhængen mellem (x, y)-koordinaterne og de nye (x_1,y_1)-koordinater, kan den kvadratiske form i de nye (x_1,y_1)-koordinater skrives på den reducerede form

$$P(x,y) = P_1(x_1,y_1) = \begin{bmatrix} x_1 & y_1 \end{bmatrix} \mathbf{Q}^{\mathrm{T}}\,\mathbf{A}\,\mathbf{Q} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix} = \begin{bmatrix} x_1 & y_1 \end{bmatrix} \begin{bmatrix} 1 & 0 \\ 0 & 3 \end{bmatrix} \begin{bmatrix} x_1 \\ y \end{bmatrix} = x_1^{\,2} + 3y_1^{\,2},$$

uden produktled af typen x_1y_1 .

Spørgsmål 2

```
> <-4|2>.Q.<x1,y1>+2;
```

$$-\sqrt{2}\,x1 + 3\sqrt{2}\,y1 + 2$$

```
> completesquare(x1^2+3*y1^2-sqrt(2)*x1+3*sqrt(2)*y1+2,[x1,y1])
  ;
```

$$3\left(y1 + \frac{\sqrt{2}}{2}\right)^2 + \left(x1 - \frac{\sqrt{2}}{2}\right)^2$$

Ved anvendelse af den positive ortogonale substitution fra forrige spørgsmål kan f skrives på den reducerede form (jævnfør ovenstående udregninger)

$$f(x,y) = f_1(x_1,y_1) = x_1^{\,2} + 3y_1^{\,2} + \begin{bmatrix} -4 & 2 \end{bmatrix} \mathbf{Q} \begin{bmatrix} x_1 \\ y_1 \end{bmatrix} + 2 = x_1^{\,2} + 3y_1^{\,2} - \sqrt{2}\,x_1 + 3\sqrt{2}\,y_1 + 2 =$$

$$\left(x_1 - \frac{\sqrt{2}}{2}\right)^2 + 3\left(y_1 + \frac{\sqrt{2}}{2}\right)^2.$$

Altså $a = 1, b = 3, c = \dfrac{\sqrt{2}}{2}$ og $d = -\dfrac{\sqrt{2}}{2}$.

Af

```
> Q.<sqrt(2)/2,-sqrt(2)/2>;
```

$$\begin{bmatrix} 1 \\ 0 \end{bmatrix}$$

følger, at $(x_1, y_1) = \left(\dfrac{\sqrt{2}}{2}, -\dfrac{\sqrt{2}}{2}\right)$ svarer til $(x, y) = (1, 0)$.

$f(1,0) = f_1\left(\dfrac{\sqrt{2}}{2}, -\dfrac{\sqrt{2}}{2}\right) = 0.$

Da

$f(x,y) - f(1,0) = f(x,y) = f_1(x_1,y_1) > 0$

for alle $(x_1,y_1) \neq \left(\dfrac{\sqrt{2}}{2}, -\dfrac{\sqrt{2}}{2}\right)$ og dermed for alle $(x, y) \neq (1, 0)$ og kun = 0

for $(x_1, y_1) = \left(\frac{\sqrt{2}}{2}, -\frac{\sqrt{2}}{2}\right)$ og dermed for $(x, y) = (1, 0)$,
så har f egentligt lokalt minimum 0 i (1, 0). Faktisk er 0 globalt minimum for f og værdimængden $f(\mathbb{R}^2) = [0; \infty[$.

Opgave 3

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
  vop:=proc(X) op(convert(X,list)) end proc:
> with(LinearAlgebra):
```

Lad a være et positivt reelt tal. En flade F i (x, y, z)-rummet givet ved parameterfremstillingen

```
> r:=(u,v)-><u^3*cos(v),u^3*sin(v),u>:
> r(u,v);
```

$$\begin{bmatrix} u^3 \cos(v) \\ u^3 \sin(v) \\ u \end{bmatrix}$$

hvor $u \in [1/2; 1]$ og $v \in [0; a]$.

Spørgsmål 1

Da den givne parameterfremstilling for fladen F har formen
$\mathbf{r}(u, v) = (g(u)\cos(v), g(u)\sin(v), h(u))$,
hvor $g(u) = u^3 > 0$ for alle $u \in [1/2; 1]$, $h(u) = u$ og $v \in [0; a]$,
så er F ifølge side 11 i eNote 25 en omdrejningsflade fremkommet ved en drejning af den i den positive (x, z)-halvplan beliggende profilkurve med parameterfremstillingen
$\mathbf{p}(u) = (u^3, 0, u)$, $u \in [1/2; 1]$,
vinklen v omkring z-aksen.

```
> plot([u^3,u,u=1/2..1],scaling=constrained,view=[0..1,0..1],
  title="Profilkurven i (x,z)-planen, der benyttes ved
  drejningen");
```



Spørgsmål 2

```
> ru:=diff(r(u,v),u);
```

$$ru := \begin{bmatrix} 3\,u^2 \cos(v) \\ 3\,u^2 \sin(v) \\ 1 \end{bmatrix}$$

```
> rv:=diff(r(u,v),v);
```

$$rv := \begin{bmatrix} -u^3 \sin(v) \\ u^3 \cos(v) \\ 0 \end{bmatrix}$$

Fladens normalvektor er

```
> N:=simplify(kryds(ru,rv));
```

$$N := \begin{bmatrix} -u^3 \cos(v) \\ -u^3 \sin(v) \\ 3\,u^5 \end{bmatrix}$$

Den til **r** hørende Jacobifunktion er

```
> Jacobi:=simplify(sqrt((prik(N,N))))assuming u>0,real;
```

$$Jacobi := u^3 \sqrt{9\,u^4 + 1}$$

Spørgsmål 3

$f(x, y, z)$ er en massetæthedsfunktion i (x, y, z)-rummet, hvor

$f(\mathbf{r}(u, v)) = \text{Jacobi}(u, v) = u^3\sqrt{1 + 9\,u^4}$ for alle $u \in [1/2; 1]$.

Den samlede masse på F er

$$\mathrm{M} = \int_F f\,\mathrm{d}\mu = \int_0^a \int_{\frac{1}{2}}^1 f(\mathbf{r}(u, v))\,\text{Jacobi}(u, v)\;\mathrm{d}u\;\mathrm{d}v = \int_0^a \int_{\frac{1}{2}}^1 \text{Jacobi}(u, v)\,\text{Jacobi}(u, v)\;\mathrm{d}u\;\mathrm{d}v$$

```
> integranden:=Jacobi*Jacobi;
```

$$integranden := u^6\left(9\,u^4 + 1\right)$$

```
> Int(Int(integranden,u=1/2..1),v=0..a)=int(int(integranden,u=
  1/2..1),v=0..a);
```

$$\int_0^a \int_{\frac{1}{2}}^1 u^6\left(9\,u^4 + 1\right)\mathrm{d}u\;\mathrm{d}v = \frac{151313\,a}{157696}$$

```
> a:=solve(151313*a*(1/157696)=1,a);
```

$$a := \frac{157696}{151313}$$

```
> evalf(a,5);
```

$$1.0422$$

Dvs F opnår massen 1 for $\mathrm{a} = \dfrac{157696}{151313} \simeq 1{,}0422.$

Opgave 4

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
  vop:=proc(X) op(convert(X,list)) end proc:
  grad:=X->convert(Student[VectorCalculus][Del](X),Vector):
  div:=V->VectorCalculus[Divergence](V):
  rot:=proc(X) uses VectorCalculus;BasisFormat(false);Curl(X)  end
  proc:
> with(LinearAlgebra):
```

En trekant T i (x, y)-planen har hjørnerne $(0,0)$, $(1, 0)$ og $(1, 2)$.

```
> P1:=plot(2*x,x=0..1,filled=true):
> P2:=plot([1,y,y=0..2]):
> display(P1,P2,scaling=constrained,title="Trekanten T");
```



Spørgsmål 1

En parameterfremstilling for T er

```
> r(u,v):=<u,v*2*u>;
```

$$r(u, v) := \begin{bmatrix} u \\ 2\,v\,u \end{bmatrix}$$

hvor $u \in [0; 1]$ og $v \in [0; 1]$

Ω er den punktmængde i (x, y, z)-rummet, som ligger lodret mellem T og planen med ligningen $z = 1$ (se figuren i opgaveteksten).

Spørgsmål 2

En parameterfremstilling for Ω er

```
> r:=(u,v,w)-><u,v*2*u,w>:
> r(u,v,w);
```

$$\begin{bmatrix} u \\ 2\,v\,u \\ w \end{bmatrix}$$

hvor $u \in [0;\,1]$, $v \in [0;\,1]$ og $w \in [0;\,1]$.

```
> M:=<diff(r(u,v,w),u)|diff(r(u,v,w),v)|diff(r(u,v,w),w)>;
```

$$M := \begin{bmatrix} 1 & 0 & 0 \\ 2\,v & 2\,u & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> Jr:=Determinant(M);
```

$$Jr := 2\,u$$

som er ≥ 0 , da $u \in [0;\,1]$. Den til **r** hørende Jacobifunktion er da

```
> Jacobi:=Jr;
```

$$Jacobi := 2\,u$$

En funktion $f := \mathbb{R}^3 \to \mathbb{R}$ er givet ved

```
> f:=(x,y,z)->x^2*y*z:
> f(x,y,z);
```

$$x^2\,y\,z$$

Vi betragter vektorfeltet $\mathbf{V}(x,\,y,\,z) = \mathbf{\nabla} f\,(x,\,y,\,z)$.

```
> V:=(x,y,z)-><diff(f(x,y,z),x),diff(f(x,y,z),y),diff(f(x,y,z),z)
  >:
> V(x,y,z);
```

$$\begin{bmatrix} 2\,x\,y\,z \\ x^2\,z \\ x^2\,y \end{bmatrix}$$

```
> divV:=div(V)(x,y,z);
```

$$divV := 2\,y\,z$$

```
> rot(V)(x,y,z);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

Spørgsmål 3

$\partial\Omega$ er den lukkede overflade af Ω med udadrettet enhedsnormalvektor. Af Gauss' sætning fås da

$$\mathrm{Flux}(\mathbf{V},\,\partial\Omega) = \int_{\Omega} \mathrm{Div}(\mathbf{V})\,\mathrm{d}\mu = \int_0^1\!\!\int_0^1\!\!\int_0^1 \mathrm{Div}(\mathbf{V})\,(\mathbf{r}(u,\,v,\,w))\,\mathrm{Jacobi}(u,\,v,\,w)\,\mathrm{d}u\,\mathrm{d}v\,\mathrm{d}w$$

```
> integranden:=div(V)(vop(r(u,v,w)))*Jacobi;
```

$$integranden := 8\, v\, u^2\, w$$

```
> Int(Int(Int(integranden,u=0..1),v=0..1),w=0..1)=int(int(int
  (integranden,u=0..1),v=0..1),w=0..1);
```

$$\int_0^1 \int_0^1 \int_0^1 8\, v\, u^2\, w \, \mathrm{d}u \, \mathrm{d}v \, \mathrm{d}w = \frac{2}{3}$$

Spørgsmål 4

En kurve K, som løber langs tre kanter på Ω fra origo til punktet P = (1, 2, 1), er vist med rødt på figuren i opgaveteksten.

Ifølge Sætning 27.10, som også gælder for en stykkevis glat kurve, er det ønskede tangentielle kurveintegral

$$\mathrm{Tan}(\nabla f, K) = \int_K \nabla f \cdot \mathbf{e} \, \mathrm{d}\mu = f(1, 2, 1) - f(0, 0\ 0) = 2 - 0 = 2.$$